# $Id$
# Maintainer: Andrea Scarpino <andrea@archlinux.org>
# Contributor: Jan de Groot <jgc@archlinux.org>
# Contributor: Tom Newsom <Jeepster@gmx.co.uk>

pkgname=libmcrypt
pkgver=2.5.8
pkgrel=4
pkgdesc="A library which provides a uniform interface to several symmetric encryption algorithms"
url="http://mcrypt.sourceforge.net/"
arch=('i686' 'x86_64')
license=('GPL')
depends=('glibc')
options=('!emptydirs')
source=("http://downloads.sourceforge.net/mcrypt/${pkgname}-${pkgver}.tar.bz2")
md5sums=('c4f491dd411a09e9de3b8702ea6f73eb')

build() {
  cd ${srcdir}/${pkgname}-${pkgver}
  ./configure --prefix=/usr --mandir=/usr/share/man
  make
}

check() {
  cd ${srcdir}/${pkgname}-${pkgver}
  make check
}

package() {
  cd ${srcdir}/${pkgname}-${pkgver}
  make DESTDIR=${pkgdir} install
}
